/*
Emily Sydnor	
Proposal title: How Conflict Orientation Influences Emotional and Political Responses to Incivility


HYPOTHESES

Stated-Hyp1: "Participants with stronger emotional responses to the treatment will be more likely to participate, either by sharing or commenting." (p. 6)

	Test-Hyp1: Higher emotional responsiveness to treatment (interaction of civility treatment x responsiveness) will increase liklihood of participation (share or comment)

Stated-Hyp2: "Conflict orientation will have an additional, indirect effect on participation through respondents' emotional responses to incivility. In other words, emotional response should serve as a mediator of the conflict orientation-participation relationship." (p. 6)

	Test-Hyp2: The positive effect of higher conflict orientation on liklihood of participation (share or comment) will be mediated by emotional responsiveness	

Stated-Hyp3: "Conflict-approaching individuals who are exposed to incivility will be more driven to participate, but will also be more likely to use uncivil discourse in their own commentary." (p. 6)

	Test-Hyp3: Conflict orientation will increase the treatment effect on liklihood of participation (share or comment)
	

Stated-Hyp4: "Conflict-avoidant respondents who are exposed to incivility will shy away from participating and will indicate that this is because of the tenor of the conversation." (p. 6)

	// the first part of this hypothesis is just the inverse of test hyp 3. 
	// the second part cannot be meaningfully tested because it requires coding 2649 text responses in an unbiased way. 
	
*/

clear all
use "TESS Sydnor_Client 5.dta", clear

********************************************************************************

* INDICATORS OF EXPERIMENTAL MANIPULATIONS

	tab XTESS216
	recode XTESS216 (1/2=0 "Civil") (3/4=1 "Uncivil"), gen(treat_uncivil)
	tab treat_uncivil

* OUTCOME MEASURES

* sharing
	/*
	Q4 [S] 
	The news organization also allows you to share the content you just watched with your social network on Facebook, Twitter, or other social networking sites. Would you be interested in sharing this story?
	*/

	tab Q4, mis
	recode Q4 (1=1) (2=0) (-1=.), gen(share)

* commenting
	/*
	Q4a [O] 
	Please write below any commentary you would include in your shared post. 
	*/
	gen nocomment = strltrim(Q4a)
	
 	tab Q4a
	gen comment=1 if nocomment!=""
	replace comment=0 if nocomment==""
	replace comment=0 if Q4a=="none" | ///
	Q4a=="n/a" | ///
	Q4a=="don't know" | ///
	Q4a=="dont know" | ///	
	Q4a=="no comments" | ///		
	Q4a=="Not sure" | ///
	Q4a=="Not at this time" | ///
	Q4a=="None" | ///		
	Q4a=="NA" | ///	
	Q4a=="N/A" | ///		
	Q4a=="N/" 
	tab comment
	tab Q4a if comment==1
		
* commented or shared
	gen participation= share
	replace participation= 1 if comment==1
	tab participation
		
* MODERATORS

* conflict orientation
	/*
	Q1 [Grid]	
	Please indicate your level of agreement with the following statements.
	Statements in rows:
	1.	I enjoy challenging the opinions of others. R
	2.	I find conflicts exciting. R
	3.	I hate arguments. 
	4.	I feel upset after an argument. R
	*/
	
	forval i=1/4 {
		tab Q1_`i'
		replace Q1_`i'=. if Q1_`i'==-1
	}
	
	// recode reverse items
	gen Q1_1r= 6-Q1_1
	gen Q1_2r= 6-Q1_2	
	gen Q1_4r= 6-Q1_4
	
	gen conflictorient = Q1_1r + Q1_2r + Q1_3 +Q1_4r
	tab conflictorient
	
* emotional reaction to the video

	/*
	To what extent did the clip you just watched make you feel any of the following?
	Statements in rows:
	1.	Angry
	2.	Disgusted
	3.	Anxious
	4.	Amused
	5.	Entertained
	6.	Enthusiastic
	*/
	
	forval i=1/4 {
		tab Q6_`i'
		replace Q6_`i'=. if Q6_`i'==-1
	}
		
	gen responsiveness = Q6_1 +Q6_2 +Q6_3 +Q6_4 +Q6_5 +Q6_6
	tab responsiveness
		
	
********************************************************************************

* ANALYSIS
	

*	Test-Hyp1: Higher emotional responsiveness to the treatment will increase liklihood of participation (share or comment)
	reg participation c.responsiveness#1.treat_uncivil
	// do not reject. 0.000
	tess c.responsiveness#1.treat_uncivil +, init(Sydnor770)
	
*	Test-Hyp2: The positive effect of higher conflict orientation on liklihood of participation (share or comment) will be mediated by emotional responsiveness	
	est sto clear
	reg participation conflictorient if treat_uncivil==1
	est sto m1
	
	reg participation conflictorient responsiveness if treat_uncivil==1
	est sto m2
	
	suest m1 m2
	test [m1_mean]conflictorient=[m2_mean]conflictorient
	// do not reject. 0.000
	tess, mediation
	
*	Test-Hyp3: Conflict orientation will increase the treatment effect on liklihood of participation (share or comment)
	reg participation c.conflictorient#i.treat_uncivil
	// do not reject. 0.000
	tess c.conflictorient#1.treat_uncivil +
	
